In[1]:=
```
SetDirectory["~/KappaLib/"];
<< kappaLib-1.2.m
<< helper.m
```

```
Loading KappaLib v1.2

Loading helper.m..
```

- **Suppose g,h are Lorentz metrics and suppose**

  Null cone(g) subset Null cone (h)

In[4]:=
```
g = DiagonalMatrix[{-1, 1, 1, 1}];
h = emMatrix["h", 4, Structure → "Symmetric"];
```

In[6]:=
```
(* define quadratic forms *)
G[v_] := v.g.v
H[v_] := v.h.v
```

- **Find null vectors for g**

In[8]:=
```
nullVectors = {
    {1, 1, 0, 0},
    {1, -1, 0, 0},

    {1, 0, 1, 0},
    {1, 0, -1, 0},

    {1, 0, 0, 1},
    {1, 0, 0, -1}

  };
```

In[9]:=
```
Table[G[nullVectors[[i]]], {i, 1, Length[nullVectors]}]
```

Out[9]=
```
{0, 0, 0, 0, 0, 0}
```

- **Since these vectors are null for g, they are null for h**

In[10]:=
```
show[Table[H[nullVectors[[i]]], {i, 1, Length[nullVectors]}]]
```

Out[10]//MatrixForm=

$$\begin{pmatrix} 1 & : & h11 + 2\,h12 + h22 \\ 2 & : & h11 - 2\,h12 + h22 \\ 3 & : & h11 + 2\,h13 + h33 \\ 4 & : & h11 - 2\,h13 + h33 \\ 5 & : & h11 + 2\,h14 + h44 \\ 6 & : & h11 - 2\,h14 + h44 \end{pmatrix}$$

- **Pairwise subtracting yields h12 = h13 = h14 = 0.**
  Moreover, h11 = -h22 = -h33 = -h44

In[11]:=
```
sub = {h12 → 0, h13 → 0, h14 → 0, h22 → -h11, h33 → -h11, h44 → -h11};
```

- **Find second set of null vectors for g**

In[12]:=
```
nullVectors = {

    {Sqrt[2], 1, 1, 0},
    {Sqrt[2], 1, 0, 1},
    {Sqrt[2], 0, 1, 1}

  };
```

In[13]:=
```
Table[G[nullVectors[[i]]], {i, 1, Length[nullVectors]}]
```

Out[13]=
```
{0, 0, 0}
```

- **Since these vectors are null for g, they are null for h**

In[14]:=
```
show[Table[H[nullVectors[[i]]], {i, 1, Length[nullVectors]}] /. sub]
```

Out[14]//MatrixForm=

$$\begin{pmatrix} 1 & : & 2\,h23 \\ 2 & : & 2\,h24 \\ 3 & : & 2\,h34 \end{pmatrix}$$

In[15]:=
```
sub = Append[sub, h23 → 0];
sub = Append[sub, h24 → 0];
sub = Append[sub, h34 → 0];
```

- **We have shown that metric h is of the form**

In[18]:=
```
h /. sub // MatrixForm
```

Out[18]//MatrixForm=

$$\begin{pmatrix} h11 & 0 & 0 & 0 \\ 0 & -h11 & 0 & 0 \\ 0 & 0 & -h11 & 0 \\ 0 & 0 & 0 & -h11 \end{pmatrix}$$

- **Since h is Lorentz, we have det (h)<0. Thus h11 != 0**

In[20]:=
```
Det[h /. sub]
```

Out[20]= $-h11^4$

- **There exists a non-zero constant Const such that**

  h = Const g

In[21]:=
```
(Const g - h) /. sub /. Const → -h11
```

Out[21]= {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}

- **The above argument follows: Richard A. Toupin, Elasticity and electro-magnetics, in: Non-Linear Continuum Theories, C.I.M.E. Conference, Bressanone, Italy 1965. C. Truesdell and G. Grioli coordinators. Pp.203-342.**